Devices Laboratory (EC 29004)

Spring - 2021

Silvaco Lab

SILVACO Atlas – a brief introduction

Dr. Gourab Dutta
Office – R207, Department of E & ECE
Email: gdutta@ece.iitkgp.ac.in

About Atlas

Atlas is a physically-based two-dimensional (2D) and three-dimensional (3D) device simulator which predicts the electrical behavior of a specified semiconductor structure. It also provides insight into the internal physical mechanisms associated with device operation.

ATLAS solves the fundamental physical equations describing the dynamics of carriers in semiconductor devices for arbitrary device Structures.

ATLAS can be used standalone or as a core tool in Silvaco's VIRTUAL WAFER FAB simulation environment.

ATLAS with other Silvaco Software

ATLAS should only be used in conjunction with the other interactive tools like DECKBUILD, TONYPLOT, DEVEDIT etc.

DECKBUILD provides an interactive run time environment.

TONYPLOT supplies scientific visualization capabilities.

DEVEDIT is an interactive tool for structure and mesh specification and refinement.

ATLAS is often used in conjunction with the ATHENA process simulator. ATHENA predicts the physical structures that result from processing steps. Electrical characteristics are predicted by ATLAS.

You can run Atlas inside the DeckBuild by using the following line:

'go atlas'



Atlas Inputs and Outputs

Most Atlas simulations use two input files-
(i) text file - contains commands for Atlas to execute
(ii) structure file - defines the structure that will be simulated



Three types of outputs produced by ATLAS:
(i) runtime output - guide to the progress of simulation that is running
(ii) log files - stores all terminal voltages and currents from the device analysis.
(iii) solution files - which stores 2D and 3D data relating to the values of solution variables within the device at a given bias point

[Athena is a process simulation tool]

General order of ATLAS Program

1. **Structure Specification** → MESH, REGION, ELECTRODE, DOPING
2. **Material Models Specification** → MATERIAL, MODELS, CONTACT, INTERFACE
3. **Numerical Method Selection** → METHOD
4. **Solution Specification** → LOG, SOLVE, LOAD, SAVE

Defining Mesh

You can define a device structure through the Atlas command language by defining the mesh. The mesh is defined by a series of horizontal and vertical lines and the spacing between them. This mesh or grid covers the physical simulation domain. The regions within this mesh can be allocated to different materials as required to construct a device.

Specifying a Mesh structure

The first statement must be:

MESH SPACE.MULT=<VALUE>

This is followed by a series of X.MESH and Y.MESH statements.

X.MESH LOCATION=<VALUE> SPACING=<VALUE>

.

Y.MESH LOCATION=<VALUE> SPACING=<VALUE>

Remember: Dimensions are in micron. Meshing should be fine in active region and coarse in bulk region.

The SPACE.MULT parameter value is used as a scaling factor for the mesh created by the X.MESH and Y.MESH statements. The default value is 1. Values **greater than 1** will create a globally **coarser mesh for fast simulation**. Values **less than 1** will create a globally **finer mesh for increased accuracy**. The X.MESH and Y.MESH statements are used to specify the locations in microns of vertical and horizontal lines, respectively, together with the vertical or horizontal spacing associated with that line. (see next page)


Tonyplot
File Edit Plot Tools Production Help
2
gate
source
drain
substrate
0
0.1
0.2
0.3
0.4
0.5
0.6
0.7
0.8
Microns
0
0.2
0.4
0.6
0.8
1
Microns

Specifying Regions and Electrodes

Once the mesh is specified, every part of it must be assigned a material type. This is done with REGION statements. For example:

REGION number=<integer> <material_type> <position parameters>
Example: *region num=1 material=GaAs y.min=0.03 y.max=0.055*

The position parameters are specified in microns using the X.MIN, X.MAX, Y.MIN, and Y.MAX parameters. (if you do not define any of the four positional parameters, values as defined in the meshing will be considered, e.g. in the above example X.MAX will be considered as the highest value of X defined during the mesh formation)

Once you have specified the regions and materials, define at least one electrode that contacts a semiconductor material. This is done with the **ELECTRODE** statement.
ELECTRODE NAME= <electrode name> <position_parameters>
Example:

Nodes that are associated with the same electrode name are treated as being electrically connected.

Specifying Doping

Doping profiles can have uniform, Gaussian, or complementary error function forms. The parameters defining the analytical distribution can be specified as:

DOPING UNIFORM CONCENTRATION=1E16 N.TYPE REGION=1

DOPING GAUSSIAN CONCENTRATION=1E18 CHARACTERISTIC=0.05 P.TYPE X.LEFT=0.0 X.RIGHT=1.0 PEAK=0.1

The first DOPING statement specifies a uniform n-type doping density of 10^{16} cm^{-3} in the region that was previously labelled as region - 1. The position parameters X.MIN, X.MAX, Y.MIN, and Y.MAX can be used instead of a region number.

The second DOPING statement specifies a p-type Gaussian profile with a peak concentration of 10^{18} cm^{-3}. This statement specifies that the peak doping is located along a line from x = 0 to x = 1 microns. Perpendicular to the peak line, the doping drops off according to a Gaussian distribution with a standard deviation of $(0.05/\sqrt{2})$ µm.

Defining other Parameters and Models

Defining Material Parameters And Models: Once the mesh, geometry, and doping profiles are defined, you can modify the characteristics of electrodes, change the default material parameters, and choose which physical models ATLAS will use during the device simulation. These actions are accomplished using the **CONTACT**, **MATERIAL**, and **MODELS** statements respectively.
Impact ionization models can be enabled using the **IMPACT** statement.
Interface properties are set by using the **INTERFACE** statement.

Work-function for Gates or Schottky Contacts. The WORKFUNCTION parameter sets the work function of the electrode.
For example: *CONTACT NAME = gate WORKFUNCTION=4.8* (specifies the work function of the gate metal)

The MATERIAL statement allows you to specify your own values for these basic parameters.
MATERIAL MATERIAL =Silicon EG300=1.12 MUN=1100 (specifies the room temp bandgap and mobility)

Specifying Physical Models: *MODELS CONMOB FLDMOB SRH FERMIDIRAC IMPACT SELB*
Specifies - Standard concentration dependent mobility (*CONMOB*) - Parallel field mobility (*FLDMOB*) - Shockley-Read-Hall recombination with fixed carrier lifetimes (*SRH*) - Fermi Dirac statistics (*FERMIDIRAC*) - Selberherr impact ionization models (*IMPACT SELB*) will be used in the simulation.

Choosing Numerical Methods

Numerical methods are given in the **METHOD** statements of the input file.

Different combinations of models will require ATLAS to solve up to six equations. For each of the model types there are basically three types of solution techniques:

(a) decoupled (GUMMEL)
(b) fully coupled (NEWTON) and
(c) BLOCK.

The **GUMMEL** method will solve for each unknown in turn keeping the other variables constant, repeating the process until a stable solution is achieved.

The **NEWTON** method solve the total system of unknowns together.

The **BLOCK** methods will solve some equations fully coupled, while others are de-coupled.

It can be useful to start a solution with a few GUMMEL iterations to generate a better guess and then switch to NEWTON to complete the solution. Specification of the solution method is carried out as follows:

METHOD GUMMEL BLOCK NEWTON

Obtaining Solutions

ATLAS can calculate DC, AC small signal, and transient solutions. Obtaining solutions is rather analogous to setting up parametric test equipment for device tests. In all simulations, the device starts with zero bias on all electrodes.

When no previous solutions exist, the initial guess for potential and carrier concentrations must be made from the doping profile. This is why the initial solution performed must be the zero bias (or thermal equilibrium) case. This is specified by the statement: *SOLVE INIT*

DC Solutions:

Sweeping The Bias:

SOLVE VCOLLECTOR=2.0
SOLVE VBASE=0.0 VSTEP=0.05 VFINAL=1.0 NAME=base

The first solution is for a collector bias of 2 V.
Next solution starts with base voltage = 0 V and runs up to 1 V with a step of 0.05 V. However, remember that during the execution of the second line the collector voltage was kept constant at 2 V.
After the execution of above lines, the collector is at 2 V and the base at 1 V.

Example

(forward I-V characteristics of a pn junction diode)

```
go atlas
mesh space.mult=1.0
x.mesh loc=0.00 spac=0.5
x.mesh loc=3.00 spac=0.2
x.mesh loc=5.00 spac=0.25
x.mesh loc=7.00 spac=0.25
x.mesh loc=9.00 spac=0.2
x.mesh loc=12.00 spac=0.5

y.mesh loc=-1.00 spac=0.1
y.mesh loc=0.00 spac=0.1
y.mesh loc=1.00 spac=0.1
y.mesh loc=2.00 spac=0.2
y.mesh loc=5.00 spac=0.4
y.mesh loc=6.00 spac=0.4

region num=1 silicon y.min=0 y.max=2.5 name=p-si
region num=2 silicon y.min=2.5 y.max=5 name=n-si
```

Start Atlas

X mesh

Y mesh



Expected structure

Defining regions

Example (cont.)

```
electrode num=1 name=anode y.min=-1 y.max=0 x.min=0 x.max=12 mat=Aluminum
electrode num=2 name=cathode y.min=5 y.max=6 x.min=0 x.max=12 mat=Aluminum
```
Electrodes

```
doping region=1 uniform p.type conc=1e17
doping region=2 uniform n.type conc=1e18
```
Defining doping

```
model conmob fldmob srh auger bgn
```
Model statement

```
output con.band val.band
```
For including band information in the structure

```
solve init
```
initial solution

```
save outf=diode_0.str
tonyplot diode_0.str -set diode_0.set
```
Save and plot the diode structure

```
method newton
```
defining numerical method

```
log outfile=diode.log
solve vanode=0.0 vstep=0.02 vfinal=1 name=anode
tonyplot diode.log -set diode_log.set
quit
```
solving for forward I-V characteristics

Example (Tonyplot output)



Structure



Mesh

Example (Tonyplot output)



Example (Tonyplot output)

Option: Tools -> Cutline or ‘F2’



Doping profile along the cutline

Right click on this plot and select ‘Display’ to see the variation of other parameters along the cutline

See Next page

Example (Tonyplot output)